

NAG Library Routine Document

G07DAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G07DAF finds the median, median absolute deviation, and a robust estimate of the standard deviation for a set of ungrouped data.

2 Specification

```
SUBROUTINE G07DAF(N, X, Y, XME, XMD, XSD, IFAIL)
```

```
INTEGER N, IFAIL
```

```
double precision X(N), Y(N), XME, XMD, XSD
```

3 Description

The data consists of a sample of size n , denoted by $x_1, x_2, \dots, x_n$, drawn from a random variable X .

G07DAF first computes the median,

$$\theta_{\text{med}} = \text{med}_i\{x_i\},$$

and from this the median absolute deviation can be computed,

$$\sigma_{\text{med}} = \text{med}_i\{|x_i - \theta_{\text{med}}|\}.$$

Finally, a robust estimate of the standard deviation is computed,

$$\sigma'_{\text{med}} = \sigma_{\text{med}} / \Phi^{-1}(0.75)$$

where $\Phi^{-1}(0.75)$ is the value of the inverse standard Normal function at the point 0.75.

G07DAF is based upon subroutine LTMDV within the ROBETH library, see Marazzi (1987).

4 References

Huber P J (1981) *Robust Statistics* Wiley

Marazzi A (1987) Subroutines for robust estimation of location and scale in ROBETH *Cah. Rech. Doc. IUMSP, No. 3 ROB 1* Institut Universitaire de Médecine Sociale et Préventive, Lausanne

5 Parameters

- | | | |
|----|---|---------------|
| 1: | N – INTEGER | <i>Input</i> |
| | <i>On entry:</i> n , the number of observations. | |
| | <i>Constraint:</i> $N > 1$. | |
| 2: | X(N) – <i>double precision</i> array | <i>Input</i> |
| | <i>On entry:</i> the vector of observations, $x_1, x_2, \dots, x_n$. | |
| 3: | Y(N) – <i>double precision</i> array | <i>Output</i> |
| | <i>On exit:</i> the observations sorted into ascending order. | |

- 4: XME – *double precision* *Output*
On exit: the median, θ_{med} .
- 5: XMD – *double precision* *Output*
On exit: the median absolute deviation, σ_{med} .
- 6: XSD – *double precision* *Output*
On exit: the robust estimate of the standard deviation, σ'_{med} .
- 7: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1 . If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0 . **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $N \leq 1$.

7 Accuracy

The computations are believed to be stable.

8 Further Comments

Unless otherwise stated in the Users' Note, the routine may be called with the same actual array supplied for parameters X and Y, in which case the sorted data values will overwrite the original contents of X. However this is not standard Fortran 77, and may not work on all systems.

9 Example

The following program reads in a set of data consisting of eleven observations of a variable X. The median, median absolute deviation and a robust estimate of the standard deviation are calculated and printed along with the sorted data in output array Y.

9.1 Program Text

```
*      G07DAF Example Program Text
*      Mark 14 Revised. NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX
      PARAMETER        (NMAX=25)
*      .. Local Scalars ..
      DOUBLE PRECISION XMD, XME, XSD
```

```

      INTEGER          I, IFAIL, N
*    .. Local Arrays ..
      DOUBLE PRECISION X(NMAX), Y(NMAX)
*    .. External Subroutines ..
      EXTERNAL         G07DAF
*    .. Executable Statements ..
      WRITE (NOUT,*) 'G07DAF Example Program Results'
*    Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      WRITE (NOUT,*)
      IF (N.LE.NMAX) THEN
         READ (NIN,*) (X(I),I=1,N)
         IFAIL = 0
*
         CALL G07DAF(N,X,Y,XME,XMD,XSD,IFAIL)
*
         WRITE (NOUT,*) 'Output Y:'
         WRITE (NOUT,99999) (Y(I),I=1,N)
         WRITE (NOUT,*)
         WRITE (NOUT,99998) 'XME = ', XME, ', XMD = ', XMD, ', XSD = ',
+           XSD
         ELSE
            WRITE (NOUT,99997) 'N is out of range: N =', N
         END IF
*
99999 FORMAT (1X,11F7.3)
99998 FORMAT (1X,A,F6.3,A,F6.3,A,F6.3)
99997 FORMAT (1X,A,I5)
      END

```

9.2 Program Data

G07DAF Example Program Data

```

11      : N, NUMBER OF OBSERVATIONS
13.0 11.0 16.0 5.0 3.0 18.0 9.0 8.0 6.0 27.0 7.0 : X, OBSERVATIONS

```

9.3 Program Results

G07DAF Example Program Results

Output Y:

```

3.000 5.000 6.000 7.000 8.000 9.000 11.000 13.000 16.000 18.000 27.000

```

```

XME = 9.000, XMD = 4.000, XSD = 5.930

```
